

Paper 368-2008

Updates to SAS® Power and Sample Size Software in SAS/STAT® 9.2

Wayne Watson, SAS Institute Inc., Cary, NC

ABSTRACT

Procedures for power analysis and sample size determination were introduced in SAS/STAT 9.1 along with a Web application for performing these analyses. SAS/STAT 9.2 adds analyses for several new designs, enhances some existing analyses, and replaces the Web application with a desktop application for your PC.

The new analyses include logistic regression; confidence intervals, equivalence tests, and noninferiority tests for a binomial proportion; and the Wilcoxon Mann-Whitney test for two distributions. These new analyses join the ones already available with SAS/STAT 9.1: general linear univariate models, one-way ANOVA, multiple regression, *t* tests, confidence intervals and equivalence tests of means, tests of proportions, and survival rank tests.

SAS Power and Sample Size (PSS) 3.1 is an application that comes with the SAS/STAT product but is installed separately. PSS 3.1 supersedes the PSS 2.0 Web application that was offered with SAS/STAT 9.1. It provides a point-and-click interface to the power analysis and sample size capabilities in SAS/STAT. Results include computation tables, graphs, and narratives (text descriptions of individual analysis scenarios). PSS 3.1 offers all of the analyses that were available in PSS 2.0 in addition to the analyses that are new for SAS 9.2. Projects that were created with PSS 2.0 can be used by PSS 3.1.

This paper highlights the new features of the procedures and describes the new desktop application.

INTRODUCTION

The POWER and GLMPOWER procedures brought power analysis and sample size determination to SAS software with SAS 9.1. PROC POWER provides power analyses for *t* tests, equivalence tests, and confidence intervals for means; exact binomial, chi-square, Fisher's exact, and McNemar tests for proportions; correlation and multiple regression; one-way ANOVA; and rank tests for comparing survival curves. PROC GLMPOWER provides analyses for linear univariate models with optional contrasts and covariates.

The SAS Power and Sample Size (PSS) 2.0 application provides easy access to the power analysis and sample size determination techniques. In addition to providing a graphical user interface for power analysis, PSS 2.0 enables you to save and reuse power analysis projects. Tables, graphs, and narratives (text descriptions) are provided as results along with the SAS statements that were generated to perform the analysis.

SAS 9.2 adds several analyses to PROC POWER, including logistic regression, rank-sum tests, and equivalence tests, noninferiority tests, and confidence intervals for one proportion. In addition, new features have been added to some of the existing analyses.

Also with SAS 9.2, PSS 3.1, a desktop application, replaces the Web application, PSS 2.0.

NEW ANALYSES

LOGISTIC REGRESSION

PROC POWER now provides power analysis for logistic regression. You can perform power and sample size analyses for the chi-square likelihood ratio test of a single predictor in a binary logistic regression, assuming independence among predictors. To perform the analysis, you choose a distribution for the predictor and specify either regression coefficients or odds ratios. You can optionally add covariates.

As an example, consider a study in which you want to investigate the presence or absence of pain symptoms as they relate to the duration of a treatment. You believe that duration is normally distributed, a reasonable odds ratio for the predictor is 1.7, and the postulated response probability (that is, the probability of the presence of pain symptoms) is 0.65. You want to calculate power for three sample sizes: 50, 60, and 70.

You use the new LOGISTIC statement to perform the power analysis for a logistic regression. The ALPHA option specifies the Type I error rate. The VARDIST option specifies the distribution of a variable (here, the predictor variable,

Duration). The distribution for treatment duration is specified as normal with a mean of 4 and a standard deviation of 1.5. The TESTPREDICTOR, TESTODDSRATIO, and RESPONSEPROB options specify the predictor variable (Duration), the odds ratio for the predictor, and the response probability, respectively. The NTOTAL option specifies the total sample sizes. Setting the POWER option to a missing value (that is, **power = .**) requests a power analysis.

```
proc power;
  logistic
    alpha = 0.05
    vardist('Duration') = normal(4, 1.5)
    testpredictor = 'Duration'
    testoddsratio = 1.7
    responseprob = 0.65
    ntotal = 50 60 70
    power = . ;
run;
```

Figure 1 shows the output, which contains two tables: the “Fixed Scenario Elements” table and the “Computed Power” table. The “Fixed Scenario Elements” table contains analysis parameters that have a single value. The “Computed Power” table contains values for parameters that have more than one value for the analysis along with the computed quantity (in this case, power). In this example, the N Total parameter has three values and is the only parameter with more than one value; all of the other parameters are constant.

Figure 1 Logistic Regression Summary Table

The POWER Procedure		
Likelihood Ratio Chi-Square Test for One Predictor		
Fixed Scenario Elements		
Method	Shieh-O'Brien approximation	
Alpha		0.05
Response Probability		0.65
Test Predictor	Duration	
Odds Ratio for Test Predictor		1.7
Unit for Test Pred Odds Ratio		1
Total Number of Bins		10
Computed Power		
Index	N Total	Power
1	50	0.664
2	60	0.742
3	70	0.805

Power is calculated for each N Total value. For the example, the resulting power for the three sample sizes ranges from 0.664 to 0.805.

As an alternative, you can specify an intercept instead of a response probability and a regression coefficient instead of an odds ratio.

If you were to add a covariate, you would need to provide its distribution in addition to either regression coefficients or odds ratios.

CONFIDENCE INTERVAL FOR ONE PROPORTION

PROC POWER now also provides analyses for the confidence interval for a binary proportion. Only two-sided intervals are supported. Six types of confidence intervals are offered; they correspond to the binomial confidence intervals computed by PROC FREQ in SAS 9.2.

Here is a simple example for the confidence interval of the proportion 0.3 with an interval width of 0.2 (half-width equals 0.1).

You use the ONESAMPLEFREQ statement with the new CI option to perform the power analysis for the confidence interval of a proportion. The PROPORTION option specifies the proportion of interest. The HALFWIDTH option specifies (half of) the width of the confidence interval. The PROBWIDTH option requests a power analysis. Prob(Width) is

calculated for confidence intervals instead of power and is slightly different than power. It is the probability of achieving the desired precision—that is, an interval with, at most, the target width.

```
proc power;
  onesamplefreq ci = Wilson
    alpha = 0.05
    proportion = 0.3
    halfwidth = 0.1
    ntotal = 70
    probwidth = .;
run;
```

As shown in [Figure 2](#), the single Prob(Width) for this example is 0.181 for a total sample size of 70.

Figure 2 Confidence Interval of One Proportion Summary Table

The POWER Procedure	
Wilson Score Confidence Interval for Binomial Proportion	
Fixed Scenario Elements	
Method	Exact
Alpha	0.05
Binomial Proportion	0.3
CI Half-Width	0.1
Total Sample Size	70
Number of Sides	2
Computed Prob(Width)	
Prob (Width)	0.181

In addition to the Wilson interval type, five other interval types are available: Wald, continuity-corrected Wald, exact, Agresti-Coull, and Jeffreys.

EQUIVALENCE TESTS FOR ONE PROPORTION

PROC POWER now provides analyses for two-sided equivalence tests for a binomial proportion. You can also perform tests of noninferiority and superiority. Exact and approximate solutions are provided. With SAS 9.2, power analysis is provided with the exact solutions, but sample size analysis is provided only with the approximate solutions.

This example is an exact equivalence test for a binomial proportion. The proportion is 0.35 and the upper and lower bounds are 0.2 and 0.4, respectively. You want to solve for power for a total sample size of 500.

You use the ONESAMPLEFREQ statement with the TEST option. You can request either an exact solution by using the **TEST = EQUIV_EXACT** option or an approximate solution by using the **TEST = EQUIV_ADJZ** option. These are new tests for SAS 9.2. The PROPORTION option specifies the proportion. The LOWER and UPPER options specify the lower and upper bounds for the equivalence test, respectively.

```
proc power;
  onesamplefreq test = equiv_exact
    alpha = 0.05
    proportion = 0.35
    lower = 0.2
    upper = 0.4
    ntotal = 500
    power = .;
run;
```

As shown in [Figure 3](#), the power for the example is 0.73 for a sample size of 500.

Figure 3 Equivalence for One Proportion Summary Table

The POWER Procedure		
Exact Test for Equivalence of Binomial Proportion		
Fixed Scenario Elements		
Method		Exact
Lower Equivalence Bound		0.2
Upper Equivalence Bound		0.4
Alpha		0.05
Binomial Proportion		0.35
Total Sample Size		500
Computed Power		
Lower Crit Val	Upper Crit Val	Power
116	181	0.730

WILCOXON MANN-WHITNEY TEST FOR TWO INDEPENDENT GROUPS

With SAS 9.2, PROC POWER provides analyses for the Wilcoxon Mann-Whitney test for two independent groups. This test is also called the Wilcoxon rank-sum test and the Mann-Whitney U test.

Here is an example from the *SAS/STAT User's Guide*. The responses to two drug combinations are being compared: lidocaine versus the combination of a fictional drug, mironel, and lidocaine. The response is a measurement of improvement, measured on a seven-point Likert scale. The scale responses range from "much worse" = -3 to "much better" = +3.

You use the new TWOSAMPLEWILCOXON statement. As in the LOGISTIC statement, you specify a distribution for a variable. In this case, there are two "variables"—one for each of the two drug combinations. In this example, the distribution is used as a way to specify that the values of the response are ordinal. The ordinal scale values (for example, -3, -2, -1, and so on) are associated with probabilities of occurrence of each ordinal value (for example, 0.01, 0.04, 0.20, and so on). The VARIABLES option specifies which groups to compare to one another. In this example, you are comparing the lidocaine group to the mironel plus lidocaine group. The SIDES option requests an upper one-sided test. The POWER option specifies the desired power. The NTOTAL option requests a sample size analysis.

```
proc power;
  twosamplewilcoxon
    alpha = 0.05
    vardist('lidocaine') =
      ordinal( (-3 -2 -1 0 1 2 3): (.01 .04 .20 .50 .20 .04 .01) )
    vardist('mironel plus lidocaine') =
      ordinal( (-3 -2 -1 0 1 2 3): (.01 .03 .15 .35 .30 .10 .06) )
    variables = 'lidocaine' | 'mironel plus lidocaine'
    sides = u
    ntotal = .
    power = 0.85;
run;
```

The output is shown in [Figure 4](#). The calculated total sample size is 176 with an actual power of 0.851. Because the calculated sample size is an integer, the actual power might differ from the requested power.

Figure 4 Wilcoxon Summary Table

The POWER Procedure		
Wilcoxon-Mann-Whitney Test		
Fixed Scenario Elements		
Method	O'Brien-Castelloe approximation	
Number of Sides		U
Alpha		0.05
Group 1 Variable		lidocaine
Group 2 Variable		mironel plus lidocaine
Nominal Power		0.85
Pooled Number of Bins		7
Group 1 Weight		1
Group 2 Weight		1
NBins Per Group		1000
Computed N Total		
	Actual	N
	Power	Total
	0.851	176

NEW FEATURES FOR EXISTING ANALYSES

PROC POWER

Several new parameterizations have been added to power analysis for paired proportions in PROC POWER. In SAS 9.1, you could specify discordant (off-diagonal) proportions or sums and differences of proportions. In SAS 9.2, you can specify the proportions in terms of p_1 and p_2 and a correlation, as well.

For two sample survival rank tests in PROC POWER, several accrual time options have been added to allow the relevant study-time portion to vary automatically with the lengths of piecewise linear survival curves.

PROC GLMPower

PROC GLMPower now supports continuous variables. It also provides a new EFFECTS option for specifying subsets of model effects and contrasts and a new ORDER option similar to the one in PROC GLM.

In this example, a full factorial model is specified, but the **EFFECTS** option requests only powers for the "A*B" interaction and the "C" main effect.

```
proc glmpower;
  class A B C;
  model Y = A B C A * B A * C B * C A * B * C;
  power
    effects = (A*B C)
    alpha = 0.05
    stddev = 2.5
    ntotal = 40.0
    power = . ;
run;
```

SAS POWER AND SAMPLE SIZE

You can also perform power analysis by using the SAS Power and Sample Size (PSS) 3.1 application. PSS 3.1 is a graphical user interface that is provided with SAS/STAT but is installed separately.

With PSS, you do not have to remember the procedure syntax. PSS prompts you for required information and validates the values that you enter. You can create table and graphical results as you can with the procedures. In addition, you can create narratives—sentence length text descriptions of power analysis scenarios. Narratives are available only in English. You can save power analysis projects and view or change them at a later time.

These features make the application very useful for occasional users. Even sophisticated users can benefit from the focus on the power analysis rather than on the SAS syntax. If you have used PSS 2.0, you no longer need a Web server to run SAS Power and Sample Size. In addition, power analysis projects that you created with PSS 2.0 can be used with PSS 3.1.

Currently, PSS is available on PCs that run the Microsoft Windows XP Professional and Vista operating systems. You can use SAS 9.2 that is installed on your PC, or you can access SAS 9.2 on another machine if SAS/CONNECT® is available on the remote machine.

To get started with SAS Power and Sample Size, select **Start►Programs►SAS►SAS Power and Sample Size►SAS Power and Sample Size 3.1**.


Here are the basic steps that you follow to use PSS.

1. Start a new project by selecting **File►New** on the menu bar or clicking the **New** icon on the toolbar.
2. In the New window, select the desired analysis type and click **OK**.

A project window for the analysis type appears with the Edit Properties page displayed. (The tabs on the Edit Properties page and their content vary according to the analysis type.)

Figure 5 shows the graphical user interface with the Edit Properties page displayed.


Figure 5 PSS Application



3. Click each tab to enter the relevant data for the analysis.
4. After you have entered all the data, click the **Calculate** button.
5. After PSS calculates the results, the project window displays the View Results page with the Summary Table tab displayed by default.

Figure 6 shows the Summary Table result. The "Computed Power" table contains powers calculated for each combination of standard deviation and total sample size. The powers range from 0.694 to 0.985. Power increases with increasing sample size per group and with a decrease in standard deviation.

Figure 6 Summary Table Result with Fixed Scenario Elements and Computed Power Tables



- To view other results or to review the SAS statements or the SAS log, click any of the tabs on the left side of the View Results page.

The graph in [Figure 7](#) shows the increase in power with sample size, but you can see that the increase in power reaches a plateau with the larger sample sizes.

Figure 7 Graph Result with Power by Sample Size Graph




Figure 8 shows a single narrative that corresponds to a sample size of 10 per group and a standard deviation of 12.

Figure 8 Narrative Result

The screenshot shows the SAS Power and Sample Size application window. The title bar reads "Experimental blood pressure drug with two groups". The window has a menu bar with "Edit Properties" and "View Results". On the left is a navigation pane with options: "Summary Table", "Graph", "Narratives" (selected), "SAS Log", and "SAS Code".

The main area is titled "Narratives" and contains the following text:

For a two-sample pooled t test of a normal mean difference with a one-sided significance level of 0.05, assuming a common standard deviation of 12, a sample size of 10 per group has a power of 0.694 to detect a difference between the means 132 and 120.

Below the narrative is a "Create Narratives" section with the instruction "Select one or more scenarios". It contains a table with the following data:

Select	Index	Sides	Alpha	Mean1	Mean2	StdDev	NPerGroup	NullDiff	Power
<input checked="" type="checkbox"/>	1	1	0.05	132	120	12	10	0	0.69355749
<input type="checkbox"/>	2	1	0.05	132	120	12	25	0	0.96713027
<input type="checkbox"/>	3	1	0.05	132	120	12	30	0	0.98545969
<input type="checkbox"/>	4	1	0.05	132	120	15	10	0	0.53038747
<input type="checkbox"/>	5	1	0.05	132	120	15	25	0	0.87355245
<input type="checkbox"/>	6	1	0.05	132	120	15	30	0	0.92176938

At the bottom of the "Create Narratives" section, there is a checkbox labeled "Hide columns with constant input values" and a button labeled "Clear all selections".

7. To print any results page, select **File►Print** on the menu bar.

CONCLUSION

The power and sample size offerings with SAS/STAT have been increased in SAS 9.2 with new analyses and new features for existing analyses in PROC POWER and GLMPower. The SAS Power and Sample Size application has been redesigned as a desktop application.

For those of you who appreciate the introductory chapters in the *SAS/STAT User's Guide*, there is now an introductory chapter for power analysis. In addition to describing the POWER and GLMPower procedures and the SAS Power and Sample Size application, it describes the use of the new %POWTABLE autocall macro for generating PROC REPORT tables of power analysis results and using the DATA step for empirical power simulation. The book, *Getting Started with the SAS Power and Sample Size Application*, has been rewritten for the desktop application and is now included as a chapter in the *SAS/STAT User's Guide* instead of being published separately.

ACKNOWLEDGMENTS

I would like to thank John Castelleo, Anne Jones, and Maura Stokes for their thoughtful reviews of the content of this paper.

CONTACT INFORMATION

Your comments and questions are valued and encouraged.

Wayne Watson
Building S, Room 3040
SAS Institute, Inc.
SAS Campus Drive
Cary, NC 27513
Work Phone: 919-531-6770
E-mail: wayne.watson@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.